
PWL_Noise_RMS
*    OUT     Rload   To create a 1V_RMS 1kHz Noise with 1Hz resolution
*     _____/\  __    Type In MacSpice Window -> rndsrc .5m   1         
*   _|_      \/  |   Get the following      
*  /   \         |   PWL_File.inc has been created in the MacSpice folder 
* /VpwlT\        |   Rename file to PWL_Noise_1K1Hz.inc     
* \     /       _|_
*  \___/        ///
*    |          Gnd     code for rndsrc is below
*   _|_
*   ///
*   Gnd      www.idea2ic.com
*            dsauersanjose@aol.com  7.15.10_3.53PM
*
*  VpwlT OUT   0  PWL(       <==  Format for piecewise linear file is as so. 
*  + 0      0.38156547         Random Noise can by generated by rndsrc 
*  + 1.00E-06 0.60066694         The step of time is 0.5ms and bandwidth
*  + 2.00E-06 0.361341313        of noise will therefore be at 1kHz. 
*  .........etc............
*  + 0.000003 0.944365605        The total measurement time is 1s which 
*  + 5.16E-04 0.658730205        corresponds to a 1Hz resolution
*  + )
*                                 This text file is located at the following 
*  Users/donsauer/Documents/MacSpice/PWL_Noise_1K1Hz.inc 
*
*   spec     start_f   stop_f   step_f         vector [vector ...]
*   spec  1      10k     1       v(out)        spec 1Hz->1kHz @ 1Hz steps
*  .tran  TSTEP  TSTOP  TSTART TMAX    
*======== ====== ====== ====== ====== ====== ====== ======
Rload     OUT    0      1k
.include                PWL_Noise_1K1Hz.inc

.control
==========Run_for_1sec_At_10KHz_BW=======================
tran      .05m   1      0     .05m     
plot       out

==========Find_Average_and_RMS=======================
let       ave = mean(out)
echo     "average  =  $&ave"
let       rms = sqrt(mean(out*out))
echo     "rms  =  $&rms"

==========Find_Spectrum=======================
linearize
set       specwindow   = "rectangular"
spec      1      10k     1       v(OUT) 
let       noise_per_Hz = 1/sqrt(1000)
plot      mag(v(OUT))    noise_per_Hz   loglog
.endc

.end

average  =  0.0654902
rms      =  0.996786

=====================END_OF_SPICE=======================

The total sample period is  1sec(1Hz)
A noise signal with 1Vrms was generated.
That corresponds to a 6Vppk signal.



Each sample point is 0.5msec apart (1KHz)
The transient is run at 50uec(10kHz) rate
The output is interpolated rather than sinc functions.
Harmonics (not Shown)are present in the spectrum. 

2
1_Vrms Noise spread our over 1000 Bins is 
reduced by a factor of 1/sqrt(1000) = 32mV_peak
Each bin has a 1Hz bandwidth.
Nyquist(-3db) is at 1KHz
Sample period is 1S (1Hz)



========="rndsrc"_File_Needs_To_Be_In_MacSpice_Folder=========

* rndsrc -- by CDHW -- writes a gausian random voltage source
*
* Note: see also the frontend command 'compose'
*
.control
begin
    setplot new
 set outfile = "PWL_File.inc"
 if ($argc = 2)
  let step = $argv[1]
  let duration =  $argv[2]
 else
  echo "usage  - rndsrc timestep duration"
  echo "effect - gaussian source written to file -- $outfile"
  unset outfile
  goto done
 endif
 
 set parity = true
 let time = 0
 echo "VpwlT OUT 0 PWL(" > $outfile
 
 while time < duration
  let time = time + step
  if $parity
    let X1 = (1+rnd(32768))/32769
    let X2 = rnd(32768)/32768*8*atan(1)
    let Vnoise = sqrt(-2*ln(X1))*cos(X2)
    set parity = false
  else
    let Vnoise = sqrt(-2*ln(X1))*sin(X2)
    set parity = true
  endif
  echo "+ $&time $&Vnoise" >> $outfile
 end
 echo "+ )" >> $outfile
 unset outfile parity
 label done
 destroy
end

echo    "PWL_File.inc has been created in the MacSpice folder"

.endc


